Some Work Scheduling Problems in *Mathematica*

The original problem

Suppose you need to staff a factory. Due to union rules all workers have to work "5 on 2 off"; that is each worker is scheduled for 5 consecutive days followed by 2 days off. You know you need to schedule 73 people for Sunday, 115 for Monday, 130 for Tuesday, 100 for Wednesday, 110 for Thursday, 150 on Friday, and 92 on Saturday. How many people should you schedule teach day to minimize your costs (assuming that all workers make the same amount)?

The Set Up

Let a be the number of people who start their 5 day work week on Sunday, b the number of people who start on Monday, c the number of people who start on Tuesday, all the way through g (the number of people who start their work week on Saturday). Then the number of people who work on Sunday is $a + d + e + f + g$, and so on. This leads to the follwing optimization problem:

Minimize $z = a + b + c + d + e + f + g$, subject to the following:

$a + d + e + f + g \geq 73$ (Sunday constraint)
$a + b + e + f + g \geq 115$ (Monday constraint)
$a + b + c + f + g \geq 130$ (Tuesday constraint)
$a + b + c + d + g \geq 100$ (Wednesday constraint)
$a + b + c + d + e \geq 110$ (Thursday constraint)
$b + c + d + e + f \geq 150$ (Friday constraint)
$c + d + e + f + g \geq 92$ (Saturday constraint)
$a, b, c, d, e, f, g \geq 0$ (non-negativity constraint)
$a, b, c, d, e, f, g \in \mathbb{Z}$ (integer values only constraint)

The Solution

We can implement the problem in *Mathematica*'s Minimize command fairly easily:

```
Clear[a, b, c, d, e, f, g]
solution1 = Minimize[
  {a + b + c + d + e + f + g, a + d + e + f + g ≥ 73 ⋀ a + b + e + f + g ≥ 115 ⋀ a + b + c + f + g ≥ 130 ⋀
    a + b + c + d + g ≥ 100 ⋀ a + b + c + d + e ≥ 110 ⋀ b + c + d + e + f ≥ 150 ⋀ c + d + e + f + g ≥ 92 ⋀
    a ≥ 0 ⋀ b ≥ 0 ⋀ c ≥ 0 ⋀ d ≥ 0 ⋀ e ≥ 0 ⋀ f ≥ 0 ⋀ g ≥ 0}, {a, b, c, d, e, f, g}, Integers]
```

```
{155, {a → 5, b → 58, c → 24, d → 13, e → 10, f → 45, g → 0}}
```

So the minimum number of workers (and therefore cost) involves 5 people starting on Sunday, 58 on Monday, 24 on Tuesday, 13 on Wednesday, 10 on Thursday, 45 on Friday, and 0 on Saturday. In terms of the number of people scheduled each day we have:

```
TableForm[ {{"Sunday", a + d + e + f + g, 73}, {"Monday", a + b + e + f + g, 115},
   {"Tuesday", a + b + c + f + g, 130 }, {"Wednesday", a + b + c + d + g, 100 },
   {"Thursday", a + b + c + d + e, 110 }, {"Friday", b + c + d + e + f, 150},
   {"Saturday", c + d + e + f + g, 92}} /. solution1[[2]],
 TableHeadings → {None, {"Day", "Actual Workers", "Minimum Requirement"}}]
```

Day	Actual Workers	Minimum Requirement
Sunday	73	73
Monday	118	115
Tuesday	132	130
Wednesday	100	100
Thursday	110	110
Friday	150	150
Saturday	92	92

In this case we can see that the “5 on, 2 off” requirement causes only a minor inefficiency - there are 3 workers more than are needed on Monday and 2 more on Tuesday.

Although the Minimize command doesn’t indicate this it turns out this problem has multiple optimal solutions. We can see this using the *Mathematica* command FindInstance and requiring that the total number of workers is 155, in this case asking for 5 solutions:

```
FindInstance[ a + b + c + d + e + f + g == 155 && a + d + e + f + g ≥ 73 ⋀
  a + b + e + f + g ≥ 115 ⋀ a + b + c + f + g ≥ 130 ⋀ a + b + c + d + g ≥ 100 ⋀
  a + b + c + d + e ≥ 110 ⋀ b + c + d + e + f ≥ 150 ⋀ c + d + e + f + g ≥ 92 ⋀ a ≥ 0 ⋀ b ≥ 0 ⋀
  c ≥ 0 ⋀ d ≥ 0 ⋀ e ≥ 0 ⋀ f ≥ 0 ⋀ g ≥ 0, {a, b, c, d, e, f, g}, Integers, 5]
```

```
{{a → 5, b → 58, c → 23, d → 17, e → 7, f → 45, g → 0},
 {a → 3, b → 57, c → 25, d → 15, e → 10, f → 44, g → 1},
 {a → 3, b → 60, c → 22, d → 14, e → 11, f → 44, g → 1},
 {a → 4, b → 58, c → 24, d → 15, e → 10, f → 43, g → 1},
 {a → 5, b → 55, c → 26, d → 14, e → 10, f → 45, g → 0}}
```

If we ask for 1000 solutions we only get 181 - so there are only 181 possible solutions:

```
Length[ FindInstance[ a + b + c + d + e + f + g == 155 && a + d + e + f + g ≥ 73 ⋀
   a + b + e + f + g ≥ 115 ⋀ a + b + c + f + g ≥ 130 ⋀ a + b + c + d + g ≥ 100 ⋀
   a + b + c + d + e ≥ 110 ⋀ b + c + d + e + f ≥ 150 ⋀ c + d + e + f + g ≥ 92 ⋀ a ≥ 0 ⋀ b ≥ 0 ⋀
   c ≥ 0 ⋀ d ≥ 0 ⋀ e ≥ 0 ⋀ f ≥ 0 ⋀ g ≥ 0, {a, b, c, d, e, f, g}, Integers, 1000]]
```

```
181
```

Something interesting happens when you remove the constraint that all the variables be integers (requiring integer solutions is what makes this type of problem computationally hard - problems with real-valued variables tend to be much easier in practice):

```
Minimize[
 {a + b + c + d + e + f + g, a + d + e + f + g ≥ 73 ⋀ a + b + e + f + g ≥ 115 ⋀ a + b + c + f + g ≥ 130 ⋀
   a + b + c + d + g ≥ 100 ⋀ a + b + c + d + e ≥ 110 ⋀ b + c + d + e + f ≥ 150 ⋀ c + d + e + f + g ≥ 92 ⋀
   a ≥ 0 ⋀ b ≥ 0 ⋀ c ≥ 0 ⋀ d ≥ 0 ⋀ e ≥ 0 ⋀ f ≥ 0 ⋀ g ≥ 0}, {a, b, c, d, e, f, g}]
```

$$\left\{\frac{463}{3}, \left\{a \to \frac{13}{3}, b \to 58, c \to \frac{70}{3}, d \to 16, e \to \frac{25}{3}, f \to \frac{133}{3}, g \to 0\right\}\right\}$$

```
N[%]
```

```
{154.333, {a → 4.33333, b → 58., c → 23.3333, d → 16., e → 8.33333, f → 44.3333, g → 0.}}
```

As you would expect you need slightly fewer workers (fewer constraints usually means a better optimal value). The suprising part is that when you round off this solution to {4, 58, 23, 16, 8, 44, 0} you don’t get any of the 181 optimal integer solutions this would take a bit of time to check by hand but we can do it quickly in *Mathematica* using the command FreeQ, which checks if an element is not in a list:

```
allsolutions =
  FindInstance[a + b + c + d + e + f + g == 155 && a + d + e + f + g ≥ 73 ⋀ a + b + e + f + g ≥ 115 ⋀
    a + b + c + f + g ≥ 130 ⋀ a + b + c + d + g ≥ 100 ⋀ a + b + c + d + e ≥ 110 ⋀
    b + c + d + e + f ≥ 150 ⋀ c + d + e + f + g ≥ 92 ⋀ a ≥ 0 ⋀ b ≥ 0 ⋀ c ≥ 0 ⋀
    d ≥ 0 ⋀ e ≥ 0 ⋀ f ≥ 0 ⋀ g ≥ 0, {a, b, c, d, e, f, g}, Integers, 1000];
FreeQ[allsolutions, {4, 58, 23, 16, 8, 44, 0}]
```

```
True
```

Mathematically this issue is important as it shows that you cannot replace the hard problem of optimizing something over the integers with the much easier problem of real-number optimization followed by rounding.

Generalizing the problem

Using *Mathematica*'s Manipulate command we can easily generalize the original/integer problem to allow for changes to the daily worker requirements (you can see the full code by opening the closed cell before the manipulation):

Sunday Requirement:	73
Monday Requirement:	115
Tuesday Requirement:	130
Wednesday Requirement:	100
Thursday Requirement:	110
Friday Requirement:	150
Saturday Requirement:	92

The total number of workers is: 155
The number starting on Sunday is: 5
The number starting on Monday is: 58
The number starting on Tuesday is: 24
The number starting on Wednesday is: 13
The number starting on Thursday is: 10
The number starting on Friday is: 45
The number starting on Saturday is: 0

Day	Actual Workers	Minimum Requirement	Excess Workers
Sunday	73	73	0
Monday	118	115	3
Tuesday	132	130	2
Wednesday	100	100	0
Thursday	110	110	0
Friday	150	150	0
Saturday	92	92	0

The total number of excess worker–days for the week is: 5

Adding in Part-Time Workers

Suppose that we alter the original problem so that 50 people are needed on Sunday, 115 on Monday, 180 on Tuesday, 100 on Wednesday, 110 on Thursday, 150 on Friday, and 92 on Saturday. Due to the large day-to-day variation the "5 on 2 off" requirement will force a large excess of workers on certain days (22 on Sunday and Monday, 38 on Wednesday, and 21 on Saturday for a total of 103 excess worker-days). One way to smooth out this excess is to allow part-time workers into the factory. Part-time workers are cheaper than full time ones (and so should contribute less cost per person to the objective function) but are not as efficient (as they can't work as many hours, accumulate experience as quickly as regular workers, have a higher turnover, etc.). Let's assume that part-time workers have to work a "4 on 3 off" schedule, contribute half as much effort as a full time worker, and cost 60% as much. How should both the full- and part-time workers be scheduled in this case?

Let the variables $a - g$ be as in the original problem. To this we add new variables to represent part-time work: h will be the number of part-time workers who start Sunday, i the number of part-time workers who start Monday, all the way through n for the number of part time workers who start on Saturday. As each part-time worker only costs 60% (or $\frac{3}{5}$) as much as a full-time worker, contributes two-thirds the labor of a full-time worker, and works for 4 consecutive days the problem will shift to:

The Set Up

Let the variables $a - g$ be as in the original problem. To this we add new variables to represent part-time work: h will be the number of part-time workers who start Sunday, i the number of part-time workers who start Monday, all the way through n for the number of part time workers who start on Saturday. As each part-time worker only costs 60% (or $\frac{3}{5}$) as much as a full-time worker, contributes two-thirds the labor of a full-time worker, and works for 4 consecutive days the problem will shift to:

Minimize $z = a + b + c + d + e + f + g + \frac{3}{5}(h + i + j + k + l + m + n)$, subject to the following:

$a + d + e + f + g + \frac{2}{3}(h + l + m + n) \geq 50$ (Sunday constraint)

$a + b + e + f + g + \frac{2}{3}(h + i + m + n) \geq 115$ (Monday constraint)

$a + b + c + f + g + \frac{2}{3}(h + i + j + n) \geq 180$ (Tuesday constraint)

$a + b + c + d + g + \frac{2}{3}(h + i + j + k) \geq 100$ (Wednesday constraint)

$a + b + c + d + e + \frac{2}{3}(i + j + k + l) \geq 110$ (Thursday constraint)

$b + c + d + e + f + \frac{2}{3}(j + k + l + m) \geq 150$ (Friday constraint)

$c + d + e + f + g + \frac{2}{3}(k + l + m + n) \geq 92$ (Saturday constraint)

$a, b, c, d, e, f, g, h, i, j, k, l, m, n \geq 0$ (non-negativity constraint)

$a, b, c, d, e, f, g, h, i, j, k, l, m, n \in \mathbb{Z}$ (integer values only constraint)

The Solution

Although we have doubled the number of variables we can still have *Mathematica* solve the problem fairly easily:

```
Clear[a, b, c, d, e, f, g, h, i, j, k, l, n]
solution2 = Minimize[ {a + b + c + d + e + f + g + 3 / 5 (h + i + j + k + l + m + n),
    a + d + e + f + g + 2 / 3 (h + l + m + n) ≥ 50 ∧ a + b + e + f + g + 2 / 3 (h + i + m + n) ≥ 115 ∧
     a + b + c + f + g + 2 / 3 (h + i + j + n) ≥ 180 ∧ a + b + c + d + g + 2 / 3 (h + i + j + k) ≥ 100 ∧
     a + b + c + d + e + 2 / 3 (i + j + k + l) ≥ 110 ∧ b + c + d + e + f + 2 / 3 (j + k + l + m) ≥ 150 ∧
     c + d + e + f + g + 2 / 3 (k + l + m + n) ≥ 92 ∧ a ≥ 0 ∧ b ≥ 0 ∧ c ≥ 0 ∧ d ≥ 0 ∧
     e ≥ 0 ∧ f ≥ 0 ∧ g ≥ 0 ∧ h ≥ 0 ∧ i ≥ 0 ∧ j ≥ 0 ∧ k ≥ 0 ∧ l ≥ 0 ∧ m ≥ 0 ∧ n ≥ 0},
   {a, b, c, d, e, f, g, h, i, j, k, l, m, n}, Integers]
```

$$\left\{\frac{864}{5}, \{a \to 0, b \to 46, c \to 22, d \to 0, e \to 0, f \to 40, g \to 0, h \to 0, i \to 0, j \to 63, k \to 0, l \to 0, m \to 0, n \to 45\}\right\}$$

This drops the payroll to $\frac{864}{5} = 172.8$ (a savings of 7.2 full-time salaries) and gives us the following schedule:

```
TableForm[ {{"Sunday", a, h}, {"Monday", b, i}, {"Tuesday", c, j}, {"Wednesday", d, k},
   {"Thursday", e, l}, {"Friday", f, m}, {"Saturday", g, n}} /. solution2[[2]],
 TableHeadings → {None, {"Start Day", "Full-time workers", "Part-time Workers"}}]
```

Start Day	Full-time workers	Part-time Workers
Sunday	0	0
Monday	46	0
Tuesday	22	63
Wednesday	0	0
Thursday	0	0
Friday	40	0
Saturday	0	45

In terms of how many people will be in the factory each day we have:

```
TableForm[
 {{"Sunday", a + d + e + f + g, h + l + m + n, N[a + d + e + f + g + 2 / 3 (h + l + m + n)], 50},
   {"Monday", a + b + e + f + g, h + i + m + n, N[a + b + e + f + g + 2 / 3 (h + i + m + n)], 115},
   {"Tuesday", a + b + c + f + g, h + i + j + n, N[a + b + c + f + g + 2 / 3 (h + i + j + n)], 180 },
   {"Wednesday", a + b + c + d + g, h + i + j + k, N[ a + b + c + d + g + 2 / 3 (h + i + j + k)], 100 },
   {"Thursday", a + b + c + d + e, i + j + k + l, N[ a + b + c + d + e + 2 / 3 (i + j + k + l)], 110 },
   {"Friday", b + c + d + e + f, j + k + l + m, N[b + c + d + e + f + 2 / 3 (j + k + l + m)], 150},
   {"Saturday", c + d + e + f + g, k + l + m + n, N[c + d + e + f + g + 2 / 3 (k + l + m + n)], 92}} /.
  solution2[[2]], TableHeadings → {None, {"Day", "Full-time Workers",
    "Part-time Workers", "Effective Full-time Workers", "Minimum Requirement"}}]
```

Day	Full-time Workers	Part-time Workers	Effective Full-time Workers
Sunday	40	45	70.
Monday	86	45	116.
Tuesday	108	108	180.
Wednesday	68	63	110.
Thursday	68	63	110.
Friday	108	63	150.
Saturday	62	45	92.

In this case there is an excess of only 31 full-time worker-days (instead of the 103 from the full-time only problem).

Generalizing the problem

Generalizing the “full-time and part-time worker” problem opens up more areas for exploration. We can of course alter the number of full-time-equivalent people who need to be at the factory each day. But more importantly we can alter the cost and relative effectiveness of the part-time workers as well:

Sunday Requirement:	50
Monday Requirement:	115
Tuesday Requirement:	180
Wednesday Requirement:	100
Thursday Requirement:	110
Friday Requirement:	150
Saturday Requirement:	92
Part–time salary (relative)	0.6
Part–time effectiveness (relative)	$\frac{2}{3}$

The total number of workers is: 108 full–time and 108 part–time

The effective number of full–time salaries is $\frac{864}{5}$

On Sunday: 0 full–time start and 0 part–time start
On Monday: 16 full–time start and 45 part–time start
On Tuesday: 23 full–time start and 63 part–time start
On Wednesday: 0 full–time start and 0 part–time start
On Thursday: 0 full–time start and 0 part–time start
On Friday: 69 full–time start and 0 part–time start
On Saturday: 0 full–time start and 0 part–time start

Day	FT Workers	PT Workers	Effective FT Workers	Minimum Requireme
Sunday	69	0	69.	50
Monday	85	45	115.015	115
Tuesday	108	108	180.036	180
Wednesday	39	108	111.036	100
Thursday	39	108	111.036	110
Friday	108	63	150.021	150
Saturday	92	0	92.	92

The total number of excess worker–days for the week is: 31.144
The savings over using full–time workers only is 7.2 full–time salaries

As you use this to explore variations of this problem keep in mind that the quantitiy to be minimized is the effective number of full-time salaries, not the number of excess worker-days (if the excess worker days are provided primarily by cheaper part-time labor the payroll might be minimized even though there’s a lot of people standing around).

Author: Dr. Christopher Moretti
Revision Date: August 3, 2015